

```


model Circuito_RC_plano
  parameter Real R = 20.0;
  parameter Real C = 0.001;
  parameter Real Vo = 10.0;
  parameter Real to = 0.05;
  Real i(start=0.0);
  Real V,Vr;
  Real u;
equation
  Vr = R*i;
  i = C*der(V);
  u - Vr = V;
  u = if time < to then 0.0 else Vo;
end Circuito_RC_plano;

```

```

model Circuito_RC_fisico
  parameter Modelica.SIunits.Resistance R = 20.0;
  parameter Modelica.SIunits.Capacitance C = 0.001;
  parameter Modelica.SIunits.Voltage Vo = 10.0;
  parameter Modelica.SIunits.Time to = 0.05;
  Modelica.SIunits.ElectricCurrent i(start=0.0);
  Modelica.SIunits.Voltage V,Vr;
  Modelica.SIunits.Voltage u;
equation
  Vr = R*i;
  i = C*der(V);
  u - Vr = V;
  u = if time < to then 0.0 else Vo;
end Circuito_RC_fisico;

```



```

model Circuito_RC_blocos
  constant Real R = 20.0;
  constant Real C = 0.001;
  constant Real Vo = 10.0;
  constant Real to = 0.05;
  Modelica.Blocks.Continuous.Integrator integrator1 annotation(
    Placement(visible = true, transformation(origin = {44, 0},
    extent = {{-10, -10}, {10, 10}}, rotation = 0)));
  Modelica.Blocks.Math.Feedback feedback1 annotation(
    Placement(visible = true, transformation(origin = {0, 0},
    extent = {{-10, -10}, {10, 10}}, rotation = 0)));
  Modelica.Blocks.Math.Gain gain1(k = 1 / R / C) annotation(
    Placement(visible = true, transformation(origin = {-36, 0},
    extent = {{-10, -10}, {10, 10}}, rotation = 0)));
  Modelica.Blocks.Math.Gain gain2(k = 1 / R / C) annotation(
    Placement(visible = true, transformation(origin = {44, -52},
    extent = {{10, -10}, {-10, 10}}, rotation = 0)));
  Modelica.Blocks.Math.Gain gain3(k = C) annotation(
    Placement(visible = true, transformation(origin = {58, 64},
    extent = {{-10, -10}, {10, 10}}, rotation = 0)));
  Modelica.Blocks.Sources.Step step1(height = Vo, startTime = to)
  annotation(
    Placement(visible = true, transformation(origin = {-82, 0},
    extent = {{-10, -10}, {10, 10}}, rotation = 0)));
  equation
    connect(gain3.u, feedback1.y) annotation(
      Line(points = {{46, 64}, {8, 64}, {8, 0}, {10, 0}}, color =
      {0, 0, 127}));
    connect(gain2.y, feedback1.u2) annotation(
      Line(points = {{32, -52}, {0, -52}, {0, -8}, {0, -8}}, color
      = {0, 0, 127}));
    connect(integrator1.y, gain2.u) annotation(
      Line(points = {{56, 0}, {80, 0}, {80, -52}, {56, -52}, {56,
      -52}}, color = {0, 0, 127}));
    connect(feedback1.y, integrator1.u) annotation(
      Line(points = {{10, 0}, {32, 0}, {32, 0}, {32, 0}}, color =
      {0, 0, 127}));
    connect(gain1.y, feedback1.u1) annotation(
      Line(points = {{-24, 0}, {-8, 0}, {-8, 0}, {-8, 0}}, color =
      {0, 0, 127}));
    connect(gain1.u, step1.y) annotation(
      Line(points = {{-48, 0}, {-72, 0}, {-72, 0}, {-70, 0}}, color
      = {0, 0, 127}));
    annotation(
      uses(Modelica(version = "3.2.2")));
end Circuito_RC_blocos;

```

```
connector pin  
  flow Real i;  
  Real v;  
end pin;
```

```
model resistor
  parameter Real R = 20.0;
  pin p, n;
equation
  R*p.i = p.v - n.v;
  p.i + n.i = 0;
end resistor;
```

```
model capacitor
  parameter Real C = 0.001;
  pin p, n;
  Real v;
equation
  v = p.v - n.v;
  p.i = C*der(v);
  p.i + n.i = 0;
end capacitor;
```



```
model terra
  pin g;
equation
  g.v = 0.0;
end terra;
```

```
model fonte_degrau
  parameter Real Vo = 10.0;
  parameter Real to = 0.05;
  pin p, n;
equation
  p.v - n.v = if time < to then 0.0 else Vo;
  p.i + n.i = 0;
end fonte_degrau;
```

```
model Circuito_RC_compontes
  resistor R(R=20.0);
  capacitor C(C=0.001);
  fonte_degrau u(Vo = 10.0, to=0.05);
  terra p;
equation
  connect(u.p,R.p);
  connect(R.n,C.p);
  connect(C.n,p.g);
  connect(p.g,u.n);
end Circuito_RC_compontes;
```

